
	Reason for change:
	1. Disconnecting and releasing the emergency voice call, as part of the f_GERAN_Postamble called in the test 11.2.8, currently does not distinguish between MO and MT case. In case of 11.2.8 which is MO call, an incorrect Transaction Identity (TI) is used in the Disconnect message, because by default TI for MT is used.
2. The 11.2.8 which is an emergency call test case without prior registration to network shares the same postamble with other non-emergency test cases. Distinction, whether the UE was or was not registered during the test and therefore whether the Detach procedure shall or shall not be part of the postamble, needs to be added in order to handle properly individual test cases if the same postamble function is to be shared.

	
	

	Summary of change:
	1. Added TI parameter to be passed from f_GERAN_Postamble to f_GERAN_CC_Disconnect functions.
2. Added Boolean parameter to f_GERAN_Postamble function to identify an emergency test case without need of detach in the postamble.

	
	

	Consequences if not approved:
	1. Conformant UE may fail this test case while rejecting the Disconnect message with cause “Invalid transaction identifier value”
2. Conformant UE fail the test by not completing the test postamble.

	Other comments:
	


Change 1
	Function Name
	f_TC_11_2_8_GERAN

	Reason for change
	1. Disconnecting and releasing the emergency voice call, as part of the f_GERAN_Postamble called in the test 11.2.8, currently does not distinguish between MO and MT case. In case of 11.2.8 which is MO call, an incorrect Transaction Identity (TI) is used in the Disconnect message, because by default TI for MT is used.

2. The 11.2.8 which is an emergency call test case without prior registration to network shares the same postamble with other non-emergency test cases. Distinction, whether the UE was or was not registered during the test and therefore whether the Detach procedure shall or shall not be part of the postamble, needs to be added in order to handle properly individual test cases if the same postamble function is to be shared.

	Summary of change
	1. Added TI parameter to be passed to f_GERAN_Postamble when called from TC 11.2.8.

2. Added boolean parameter to be passed to f_GERAN_Postamble when called from TC 11.2.8.

	Source of change
	LTE_IRAT\11\IMS_EmergencyCall_UGC_GERAN.ttcn

	MCC160
	


Before:
	 function f_TC_11_2_8_GERAN() runs on GERAN_PTC

  {

    var G_LogicChType v_TchType;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell(geran_Cell24);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell

    f_GERAN_CreateTCH(geran_Cell24);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_ServingCell);

    f_GERAN_SetChPwrLevel(geran_Cell24, tsc_TchCh, tsc_GERAN_ServingCell);

    //@siclog "Step 4b1 - 4b7" siclog@

    v_TchType := f_GERAN_EmgCallSetup(geran_Cell24);

    f_GERAN_TestBody_Set(false);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);     // Tell EUTRA test is over

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G4_CSCALL, -, -, v_TchType);
  } // f_TC_11_2_8_GERAN




After:
	import from NAS_24008Templates_IRAT all;

  function f_TC_11_2_8_GERAN() runs on GERAN_PTC

  {

    var G_LogicChType v_TchType;

    //Initialise cell info and configure cell 24

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell(geran_Cell24);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    //Receive updated auth parameters

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell

    f_GERAN_CreateTCH(geran_Cell24);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell24, tsc_GERAN_ServingCell);

    f_GERAN_SetChPwrLevel(geran_Cell24, tsc_TchCh, tsc_GERAN_ServingCell);

    //@siclog "Step 4b1 - 4b7" siclog@

    v_TchType := f_GERAN_EmgCallSetup(geran_Cell24);

    f_GERAN_TestBody_Set(false);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);     // Tell EUTRA test is over

    //Perform postamble

    f_GERAN_Postamble(geran_Cell24, G4_CSCALL, -, -, v_TchType, cs_TI_MO, true); //Anritsu TTCN CR 11.2.8

  } // f_TC_11_2_8_GERAN


Change 2
	Function Name
	f_GERAN_Postamble

	Reason for change
	Changes corresponding to those in Change 1 need to be reflected in the definition of f_GERAN_Postamble  function.mble.y individual test cases with other non-e

	Summary of change
	TI parameter is now passed to f_GERAN_CC_Disconnect function and execution of f_GERAN_SwitchPowerOff now depends on the new Boolean parameter received from f_GERAN_Postamble function in case of G4_CSCALL state.

	Source of change
	LTE_IRAT\11\IMS_EmergencyCall_UGC_GERAN.ttcn

	MCC160
	


Before:
	  function f_GERAN_Postamble(GERAN_CellId_Type         p_CellId,

                             GERAN_POSTAMBLE_STATE_TYPE p_GERAN_POSTAMBLE_STATE,

                             UTRAN_GERAN_ATTACH_Type p_AttachType := NotSpecified,

                             boolean p_DoUpdate := false,

                             G_LogicChType p_TCHType := -1) runs on GERAN_PTC

  {

    var boolean v_IsCombined := f_GERAN_AttachTypeCheck(p_AttachType); // @sic R5s110176 sic@

    var B2_Type v_Nmo := f_GERAN_NMO_Get(p_CellId);

    var B3_Type v_UpdateType := tsc_I_UpdateTypeRAOnly;

    var boolean v_IsTCH := false;

    if (p_TCHType > -1) { // @sic R5s120498 sic@

      v_IsTCH := true;

    }

    if (v_IsCombined and v_Nmo == '00'B) {

      v_UpdateType := tsc_I_UpdateTypeCombined;

    }

    select (p_GERAN_POSTAMBLE_STATE) {

      case (G1_IDLE) {

        //Switch/Power off UE

        f_GERAN_SwitchPowerOff(p_CellId, p_AttachType); // @sic R5s110176 sic@

      }

      case (G2_PSHANDOVER) {

        if (p_DoUpdate) {

          f_GERAN_RAU(p_CellId, '0'B, v_UpdateType);

          if (v_IsCombined and v_Nmo == '01'B) { // If NMO II and C2, then have to do LAU too

            f_GERAN_LocationUpdate(p_CellId);

          }

        }

        //Switch/Power off UE

        f_GERAN_SwitchPowerOff(p_CellId, p_AttachType); // @sic R5s110176 sic@

      }

      case (G3_CSFALLBACK) {

        //Perform CS Fallback procedure

        f_GERAN_CS_Fallback(p_CellId, p_AttachType);

        //Switch/Power off UE

        f_GERAN_SwitchPowerOff(p_CellId, p_AttachType); // @sic R5s110176 sic@

      }

      case (G4_CSCALL) {

        //Disconnect the active CC call

        f_GERAN_CC_Disconnect(p_CellId, p_TCHType);
        v_IsTCH := true; // @sic R5s120498 sic@

        //SwitchPower off UE

        f_GERAN_SwitchPowerOff(p_CellId, p_AttachType); // @sic R5s110176 sic@

      }

      case (G5_DEREGISTERED) {

        if (pc_SwitchOnOff == true) {

          //Simply switch the UE off

          f_UT_SwitchOffUE(UT, false);

        } else {//just remove power from UE

          f_UT_PowerOffUE(UT);

        }

      }

      case else {//No state defined - just remove power from UE

        f_UT_PowerOffUE(UT);

      }

    }

    //Release the GERAN cell

    f_GERAN_SS_CellRelease(p_CellId, -, v_IsTCH); // @sic R5s120498 sic@

  }




After:
	function f_GERAN_Postamble(GERAN_CellId_Type         p_CellId,

                             GERAN_POSTAMBLE_STATE_TYPE p_GERAN_POSTAMBLE_STATE,

                             UTRAN_GERAN_ATTACH_Type p_AttachType := NotSpecified,

                             boolean p_DoUpdate := false,

                             G_LogicChType p_TCHType := -1,

                             template (value) TI p_Calltype := cs_TI_MT,

                             boolean p_Emergency := false) runs on GERAN_PTC //Anritsu TTCN CR 11.2.8

  {

    var boolean v_IsCombined := f_GERAN_AttachTypeCheck(p_AttachType); // @sic R5s110176 sic@

    var B2_Type v_Nmo := f_GERAN_NMO_Get(p_CellId);

    var B3_Type v_UpdateType := tsc_I_UpdateTypeRAOnly;

    var boolean v_IsTCH := false;

    if (p_TCHType > -1) { // @sic R5s120498 sic@

      v_IsTCH := true;

    }

    if (v_IsCombined and v_Nmo == '00'B) {

      v_UpdateType := tsc_I_UpdateTypeCombined;

    }

    select (p_GERAN_POSTAMBLE_STATE) {

      case (G1_IDLE) {

        //Switch/Power off UE

        f_GERAN_SwitchPowerOff(p_CellId, p_AttachType); // @sic R5s110176 sic@

      }

      case (G2_PSHANDOVER) {

        if (p_DoUpdate) {

          f_GERAN_RAU(p_CellId, '0'B, v_UpdateType);

          if (v_IsCombined and v_Nmo == '01'B) { // If NMO II and C2, then have to do LAU too

            f_GERAN_LocationUpdate(p_CellId);

          }

        }

        //Switch/Power off UE

        f_GERAN_SwitchPowerOff(p_CellId, p_AttachType); // @sic R5s110176 sic@

      }

      case (G3_CSFALLBACK) {

        //Perform CS Fallback procedure

        f_GERAN_CS_Fallback(p_CellId, p_AttachType);

        //Switch/Power off UE

        f_GERAN_SwitchPowerOff(p_CellId, p_AttachType); // @sic R5s110176 sic@

      }

      case (G4_CSCALL) {

        //Disconnect the active CC call

        f_GERAN_CC_Disconnect(p_CellId, p_TCHType, -, p_Calltype); //Anritsu TTCN CR 11.2.8

        v_IsTCH := true; // @sic R5s120498 sic@

        if (not(p_Emergency)) {

        //SwitchPower off UE

        f_GERAN_SwitchPowerOff(p_CellId, p_AttachType); // @sic R5s110176 sic@

        }
      }

      case (G5_DEREGISTERED) {

        if (pc_SwitchOnOff == true) {

          //Simply switch the UE off

          f_UT_SwitchOffUE(UT, false);

        } else {//just remove power from UE

          f_UT_PowerOffUE(UT);

        }

      }

      case else {//No state defined - just remove power from UE

        f_UT_PowerOffUE(UT);

      }

    }

    //Release the GERAN cell

    f_GERAN_SS_CellRelease(p_CellId, -, v_IsTCH); // @sic R5s120498 sic@

  }


